

Sauer-Shelah Lemma

Ata Keskin

November 18, 2022

Abstract

The Sauer-Shelah Lemma is a fundamental result in extremal set theory and combinatorics, that guarantees the existence of a set T of size k which is shattered by a family of sets \mathcal{F} , if the cardinality of the family is greater than some bound dependent on k . A set T is said to be shattered by a family \mathcal{F} if every subset of T can be obtained as an intersection of T with some set $S \in \mathcal{F}$. The Sauer-Shelah Lemma has found use in diverse fields such as computational geometry, approximation algorithms and machine learning. In this entry we formalize the notion of shattering and prove the generalized and standard versions of the Sauer-Shelah Lemma.

Contents

1	Introduction	1
2	Definitions and lemmas about shattering	2
2.1	Intersection of a family of sets with a set	2
2.2	Definition of shattering	3
3	Lemmas involving the cardinality of sets	5
4	Lemmas involving the binomial coefficient	7
5	Sauer-Shelah Lemma	8
5.1	Generalized Sauer-Shelah Lemma	8
5.2	Sauer-Shelah Lemma	11
5.3	Sauer-Shelah Lemma for hypergraphs	11
5.4	Alternative statement of the Sauer-Shelah Lemma	11

1 Introduction

The goal of this entry is to formalize the Sauer-Shelah Lemma. The result was first published by Sauer [2] and Shelah [3] independent from one another. The proof presented in this entry is based on an article by Kalai [1].

The lemma has a wide range of applications. Vapnik and Chervonenkis [4] reproved and used the lemma in the context of statistical learning theory. For instance, the VC-dimension of a family of sets is defined as the size of the largest set the family shatters. In this context the Sauer-Shelah Lemma is a result tying the VC-dimension of a family to the number of sets in the family.

2 Definitions and lemmas about shattering

```
theory Shattering
  imports Main
begin
```

2.1 Intersection of a family of sets with a set

```
abbreviation IntF :: 'a set set  $\Rightarrow$  'a set  $\Rightarrow$  'a set set (infixl  $\cap^*$  60)
  where  $F \cap^* S \equiv ((\cap) S) ' F$ 
```

```
lemma idem-IntF:
  assumes  $\bigcup A \subseteq Y$ 
  shows  $A \cap^* Y = A$ 
proof -
  from assms have  $A \subseteq A \cap^* Y$  by blast
  thus ?thesis by fastforce
qed
```

```
lemma subset-IntF:
  assumes  $A \subseteq B$ 
  shows  $A \cap^* X \subseteq B \cap^* X$ 
  using assms by (rule image-mono)
```

```
lemma Int-IntF:  $(A \cap^* Y) \cap^* X = A \cap^* (Y \cap X)$ 
proof
  show  $A \cap^* Y \cap^* X \subseteq A \cap^* (Y \cap X)$ 
  proof
    fix  $S$ 
    assume  $S \in A \cap^* Y \cap^* X$ 
    then obtain a-y where A-Y0:  $a-y \in A \cap^* Y$  and A-Y1:  $a-y \cap X = S$  by
      blast
    from A-Y0 obtain a where A0:  $a \in A$  and A1:  $a \cap Y = a-y$  by blast
    from A-Y1 A1 have  $a \cap (Y \cap X) = S$  by fast
    with A0 show  $S \in A \cap^* (Y \cap X)$  by blast
  qed
next
  show  $A \cap^* (Y \cap X) \subseteq A \cap^* Y \cap^* X$ 
  proof
    fix  $S$ 
    assume  $S \in A \cap^* (Y \cap X)$ 
```

then obtain a where $A0: a \in A$ and $A1: a \cap (Y \cap X) = S$ by *blast*
 from $A0$ have $a \cap Y \in A \cap* Y$ by *blast*
 with $A1$ show $S \in (A \cap* Y) \cap* X$ by *blast*
 qed
 qed
 insert distributes over IntF
lemma *insert-IntF*:
 shows $\text{insert } x \text{ ' } (H \cap* S) = (\text{insert } x \text{ ' } H) \cap* (\text{insert } x \text{ ' } S)$
proof
 show $\text{insert } x \text{ ' } (H \cap* S) \subseteq (\text{insert } x \text{ ' } H) \cap* (\text{insert } x \text{ ' } S)$
proof
 fix $y-x$
 assume $y-x \in \text{insert } x \text{ ' } (H \cap* S)$
 then obtain y where $0: y \in (H \cap* S)$ and $1: y-x = y \cup \{x\}$ by *blast*
 from 0 obtain yh where $2: yh \in H$ and $3: y = yh \cap S$ by *blast*
 from $1 \ 3$ have $y-x = (yh \cup \{x\}) \cap (S \cup \{x\})$ by *simp*
 with 2 show $y-x \in (\text{insert } x \text{ ' } H) \cap* (\text{insert } x \text{ ' } S)$ by *blast*
 qed
next
 show $\text{insert } x \text{ ' } H \cap* (\text{insert } x \text{ ' } S) \subseteq \text{insert } x \text{ ' } (H \cap* S)$
proof
 fix $y-x$
 assume $y-x \in \text{insert } x \text{ ' } H \cap* (\text{insert } x \text{ ' } S)$
 then obtain $yh-x$ where $0: yh-x \in (\lambda Y. Y \cup \{x\}) \text{ ' } H$ and $1: y-x = yh-x \cap (S \cup \{x\})$ by *blast*
 from 0 obtain yh where $2: yh \in H$ and $3: yh-x = yh \cup \{x\}$ by *blast*
 from $1 \ 3$ have $y-x = (yh \cap S) \cup \{x\}$ by *simp*
 with 2 show $y-x \in \text{insert } x \text{ ' } (H \cap* S)$ by *blast*
 qed
 qed

2.2 Definition of shattering

abbreviation *shatters* :: 'a set set \Rightarrow 'a set \Rightarrow bool (*infixl shatters 70*)
 where $H \text{ shatters } A \equiv H \cap* A = \text{Pow } A$

definition *VC-dim* :: 'a set set \Rightarrow nat
 where $\text{VC-dim } F = \text{Sup } \{\text{card } S \mid S. F \text{ shatters } S\}$

definition *shattered-by* :: 'a set set \Rightarrow 'a set set
 where $\text{shattered-by } F \equiv \{A. F \text{ shatters } A\}$

lemma *shattered-by-in-Pow*:
 shows $\text{shattered-by } F \subseteq \text{Pow } (\bigcup F)$
 unfolding *shattered-by-def* by *blast*

lemma *subset-shatters*:
 assumes $A \subseteq B$ and $A \text{ shatters } X$
 shows $B \text{ shatters } X$

```

proof –
  from assms(1) have  $A \cap^* X \subseteq B \cap^* X$  by blast
  with assms(2) have  $Pow\ X \subseteq B \cap^* X$  by presburger
  thus ?thesis by blast
qed

lemma supset-shatters:
  assumes  $Y \subseteq X$  and  $A$  shatters  $X$ 
  shows  $A$  shatters  $Y$ 
proof –
  have  $h: \bigcup (Pow\ Y) \subseteq Y$  by simp
  from assms have  $0: Pow\ Y \subseteq A \cap^* X$  by auto
  from subset-IntF[OF 0, of  $Y$ ] Int-IntF[of  $Y\ X\ A$ ] idem-IntF[OF  $h$ ] have  $Pow\ Y \subseteq A \cap^* (X \cap Y)$  by argo
  with Int-absorb2[OF assms(1)] Int-commute[of  $X\ Y$ ] have  $Pow\ Y \subseteq A \cap^* Y$ 
by presburger
  then show ?thesis by fast
qed

lemma shatters-empty:
  assumes  $F \neq \{\}$ 
  shows  $F$  shatters  $\{\}$ 
using assms by fastforce

lemma subset-shattered-by:
  assumes  $A \subseteq B$ 
  shows shattered-by  $A \subseteq$  shattered-by  $B$ 
unfolding shattered-by-def using subset-shatters[OF assms] by force

lemma finite-shattered-by:
  assumes finite  $(\bigcup\ F)$ 
  shows finite (shattered-by  $F$ )
  using assms rev-finite-subset[OF - shattered-by-in-Pow, of  $F$ ] by fast

  The following example shows that requiring finiteness of a family of sets
  is not enough

lemma  $\exists F::nat\ set\ set. finite\ F \wedge infinite\ (shattered-by\ F)$ 
proof –
  let  $?F = \{odd - ' \{True\}, odd - ' \{False\}\}$ 
  have  $0: finite\ ?F$  by simp

  let  $?f = \lambda n::nat. \{n\}$ 
  let  $?N = range\ ?f$ 
  have inj  $(\lambda n. \{n\})$  by simp
  with infinite-iff-countable-subset[of  $?N$ ] have infinite-N: infinite  $?N$  by blast
  have F-shatters-any-singleton:  $?F$  shatters  $\{n::nat\}$  for  $n$ 
  proof –
    have  $Pow\ n: Pow\ \{n\} = \{\{n\}, \{\}\}$  by blast
    have  $1: Pow\ \{n\} \subseteq ?F \cap^* \{n\}$ 

```

```

proof (cases odd n)
  case True
    from True have (odd - ' {False} )  $\cap$  {n} = {} by blast
    hence 0: {}  $\in$  ?F  $\cap$ * {n} by blast
    from True have (odd - ' {True} )  $\cap$  {n} = {n} by blast
    hence 1: {n}  $\in$  ?F  $\cap$ * {n} by blast
    from 0 1 Pow-n show ?thesis by simp
  next
    case False
    from False have (odd - ' {True} )  $\cap$  {n} = {} by blast
    hence 0: {}  $\in$  ?F  $\cap$ * {n} by blast
    from False have (odd - ' {False} )  $\cap$  {n} = {n} by blast
    hence 1: {n}  $\in$  ?F  $\cap$ * {n} by blast
    from 0 1 Pow-n show ?thesis by simp
  qed
  thus ?thesis by fastforce
qed
then have ?N  $\subseteq$  shattered-by ?F unfolding shattered-by-def by force
from 0 infinite-super[OF this infinite-N] show ?thesis by blast
qed

```

end

3 Lemmas involving the cardinality of sets

theory *Card-Lemmas*

imports *Main*

begin

lemma *card-diff*:

assumes *finite A*

shows *card A* = *card (A - B)* + *card (A \cap B)*

proof -

from *assms* **have** *fin0: finite (A - B)* **and** *fin1: finite (A \cap B)* **by** *blast+*

have *A-equ: A = (A - B) \cup (A \cap B)* **and** *disjoint: (A - B) \cap (A \cap B) = {}*

by *blast+*

from *card-Un-disjoint*[*OF fin0 fin1 disjoint*] *A-equ* **show** ?*thesis* **by** *argo*

qed

lemma *card-Int-copy*:

assumes *finite X* **and** *A \cup B \subseteq X* **and** $\exists f. \text{inj-on } f \text{ } (A \cap B) \wedge (A \cup B) \cap (f \text{ ` } (A \cap B)) = \{\}$

shows *card A* + *card B* \leq *card X*

proof -

from *rev-finite-subset*[*OF assms(1), of A*] *rev-finite-subset*[*OF assms(1), of B*] *assms(2)*

have *finite-A: finite A* **and** *finite-B: finite B* **by** *blast+*

then **have** *finite-A-Un-B: finite (A \cup B)* **and** *finite-A-Int-B: finite (A \cap B)* **by**

blast+
from $\text{assms}(3)$ **obtain** f **where** $f\text{-inj-on}$: $\text{inj-on } f \ (A \cap B)$ **and** $f\text{-disjnt}$: $(A \cup B) \cap (f \text{ ` } (A \cap B)) = \{\}$ **and** $f\text{-inj-in}$: $f \text{ ` } (A \cap B) \subseteq X$ **by** blast
from finite-A-Int-B **have** finite-f-img : $\text{finite } (f \text{ ` } (A \cap B))$ **by** blast
from $\text{assms}(2)$ $f\text{-inj-in}$ **have** union-in : $(A \cup B) \cup f \text{ ` } (A \cap B) \subseteq X$ **by** blast

from $\text{card-Un-Int}[OF \ \text{finite-A} \ \text{finite-B}]$ **have** $\text{card } A + \text{card } B = \text{card } (A \cup B) + \text{card } (A \cap B)$.
also from $\text{card-image}[OF \ f\text{-inj-on}]$ **have** $\dots = \text{card } (A \cup B) + \text{card } (f \text{ ` } (A \cap B))$ **by** presburger
also from $\text{card-Un-disjoint}[OF \ \text{finite-A-Un-B} \ \text{finite-f-img} \ f\text{-disjnt}]$ **have** $\dots = \text{card } ((A \cup B) \cup f \text{ ` } (A \cap B))$ **by** argo
also from $\text{card-mono}[OF \ \text{assms}(1) \ \text{union-in}]$ **have** $\dots \leq \text{card } X$ **by** blast
finally show $?thesis$.
qed

lemma card-ge-0 :
assumes $A \neq \{\}$ **and** $\text{finite } A$
shows $0 < \text{card } A$
proof –
from $\text{assms}(1)$ **have** $\{\} \subset A$ **by** blast
from $\text{psubset-card-mono}[OF \ \text{assms}(2) \ \text{this}]$ **show** $?thesis$ **by** force
qed

lemma $\text{finite-diff-not-empty}$:
assumes $\text{finite } Y$ **and** $\text{card } Y < \text{card } X$
shows $X - Y \neq \{\}$
proof
assume $X - Y = \{\}$
hence $X \subseteq Y$ **by** simp
from $\text{card-mono}[OF \ \text{assms}(1) \ \text{this}]$ $\text{assms}(2)$ **show** False **by** linarith
qed

lemma $\text{obtain-difference-element}$:
fixes $F :: 'a \text{ set set}$
assumes $2 \leq \text{card } F$
obtains x **where** $x \in \bigcup F$ $x \notin \bigcap F$
proof –
from assms $\text{card-le-Suc-iff}[of \ 1 \ F]$ **obtain** $A \ F'$ **where** $0: F = \text{insert } A \ F'$ **and** $1: A \notin F'$ **and** $2: 1 \leq \text{card } F'$ **by** auto
from 2 $\text{card-le-Suc-iff}[of \ 0 \ F]$ **obtain** $B \ F''$ **where** $3: F' = \text{insert } B \ F''$ **by** auto
from $1 \ 3$ **have** $A\text{-noteq-B}$: $A \neq B$ **by** blast
from $0 \ 3$ **have** $A\text{-in-F}$: $A \in F$ **and** $B\text{-in-F}$: $B \in F$ **by** blast+
from $A\text{-noteq-B}$ **have** $(A - B) \cup (B - A) \neq \{\}$ **by** simp
with $A\text{-in-F} \ B\text{-in-F}$ **that** **show** thesis **by** blast
qed

end

4 Lemmas involving the binomial coefficient

theory *Binomial-Lemmas*

imports *Main*

begin

lemma *choose-mono*:

assumes $x \leq y$

shows $x \text{ choose } n \leq y \text{ choose } n$

proof –

have *finite* $\{0..<y\}$ **by** *blast*

with *finite-Pow-iff*[*of* $\{0..<y\}$] **have** *finiteness*: *finite* $\{K \in \text{Pow } \{0..<y\}. \text{card } K = n\}$ **by** *simp*

from *assms* **have** $\text{Pow } \{0..<x\} \subseteq \text{Pow } \{0..<y\}$ **by** *force*

then **have** $\{K \in \text{Pow } \{0..<x\}. \text{card } K = n\} \subseteq \{K \in \text{Pow } \{0..<y\}. \text{card } K = n\}$ **by** *blast*

from *card-mono*[*OF finiteness this*] **show** *?thesis* **unfolding** *binomial-def* .

qed

lemma *choose-row-sum-set*:

assumes *finite* $(\bigcup F)$

shows $\text{card } \{S. S \subseteq \bigcup F \wedge \text{card } S \leq k\} = (\sum i \leq k. \text{card } (\bigcup F) \text{ choose } i)$

proof (*induction k*)

case 0

from *rev-finite-subset*[*OF assms*] **have** $S \subseteq \bigcup F \wedge \text{card } S \leq 0 \longleftrightarrow S = \{\}$ **for** S **by** *fastforce*

then **show** *?case* **by** *simp*

next

case (*Suc k*)

let $?FS = \{S. S \subseteq \bigcup F \wedge \text{card } S \leq \text{Suc } k\}$

and $?F\text{-}Asm = \{S. S \subseteq \bigcup F \wedge \text{card } S \leq k\}$

and $?F\text{-}Step = \{S. S \subseteq \bigcup F \wedge \text{card } S = \text{Suc } k\}$

from *finite-Pow-iff*[*of* $\bigcup F$] *assms* **have** *finite-Pow-Un-F*: *finite* $(\text{Pow } (\bigcup F))$..

have $?F\text{-}Asm \subseteq \text{Pow } (\bigcup F)$ **and** $?F\text{-}Step \subseteq \text{Pow } (\bigcup F)$ **by** *fast+*

with *rev-finite-subset*[*OF finite-Pow-Un-F*] **have** *finite-F-Asm*: *finite* $?F\text{-}Asm$ **and** *finite-F-Step*: *finite* $?F\text{-}Step$ **by** *presburger+*

have $F\text{-}Un$: $?FS = ?F\text{-}Asm \cup ?F\text{-}Step$ **and** $F\text{-}disjoint$: $?F\text{-}Asm \cap ?F\text{-}Step = \{\}$ **by** *fastforce+*

from *card-Un-disjoint*[*OF finite-F-Asm finite-F-Step F-disjoint*] $F\text{-}Un$ **have** $\text{card } ?FS = \text{card } ?F\text{-}Asm + \text{card } ?F\text{-}Step$ **by** *argo*

also **from** *Suc* **have** $\dots = (\sum i \leq k. \text{card } (\bigcup F) \text{ choose } i) + \text{card } ?F\text{-}Step$ **by** *argo*

also **from** *n-subsets*[*OF assms, of Suc k*] **have** $\dots = (\sum i \leq \text{Suc } k. \text{card } (\bigcup F) \text{ choose } i)$ **by** *force*

finally **show** *?case* **by** *blast*

qed

end

5 Sauer-Shelah Lemma

```

theory Sauer-Shelah-Lemma
  imports Main Shattering Card-Lemmas Binomial-Lemmas
begin

```

5.1 Generalized Sauer-Shelah Lemma

```

lemma sauer-shelah-0:
  fixes  $F :: 'a \text{ set set}$ 
  shows  $\text{finite } (\bigcup F) \implies \text{card } F \leq \text{card } (\text{shattered-by } F)$ 
proof (induction  $F$  rule: measure-induct-rule[of card])
  case (less  $F$ )
  note  $\text{finite-}F = \text{finite-UnionD}[OF \text{ less}(2)]$ 
  note  $\text{finite-sh}F = \text{finite-shattered-by}[OF \text{ less}(2)]$ 
  show ?case
  proof (cases  $2 \leq \text{card } F$ )
  case True
    from obtain-difference-element[OF True] obtain  $x :: 'a$  where  $x\text{-in-Union-}F$ :
 $x \in \bigcup F$  and  $x\text{-not-in-Int-}F$ :  $x \notin \bigcap F$  by blast

```

Define F_0 as the subfamily of F containing those sets that don't contain

x

```

  let ? $F_0 = \{S \in F. x \notin S\}$ 
  from  $x\text{-in-Union-}F$  have  $F_0\text{-psubset-}F$ : ? $F_0 \subset F$  by blast
  from  $F_0\text{-psubset-}F$  have  $F_0\text{-in-}F$ : ? $F_0 \subseteq F$  by blast
  from subset-shattered-by[OF  $F_0\text{-in-}F$ ] have  $\text{sh}F_0\text{-subset-sh}F$ :  $\text{shattered-by } ?F_0$ 
 $\subseteq \text{shattered-by } F$  .
  from  $F_0\text{-in-}F$  have  $Un\text{-}F_0\text{-in-}Un\text{-}F$ :  $\bigcup ?F_0 \subseteq \bigcup F$  by blast

```

F_0 shatters at least as many sets as $|F_0|$ by the induction hypothesis

```

  note  $IH\text{-}F_0 = \text{less}(1)[OF \text{ psubset-card-mono}[OF \text{ finite-}F \text{ } F_0\text{-psubset-}F] \text{ rev-finite-subset}[OF$ 
 $\text{less}(2) \text{ } Un\text{-}F_0\text{-in-}Un\text{-}F]]$ 

```

Define F_1 as the subfamily of F containing those sets that contain x

```

  let ? $F_1 = \{S \in F. x \in S\}$ 
  from  $x\text{-not-in-Int-}F$  have  $F_1\text{-psubset-}F$ : ? $F_1 \subset F$  by blast
  from  $F_1\text{-psubset-}F$  have  $F_1\text{-in-}F$ : ? $F_1 \subseteq F$  by blast
  from subset-shattered-by[OF  $F_1\text{-in-}F$ ] have  $\text{sh}F_1\text{-subset-sh}F$ :  $\text{shattered-by } ?F_1$ 
 $\subseteq \text{shattered-by } F$  .
  from  $F_1\text{-in-}F$  have  $Un\text{-}F_1\text{-in-}Un\text{-}F$ :  $\bigcup ?F_1 \subseteq \bigcup F$  by blast

```

F_1 shatters at least as many sets as $|F_1|$ by the induction hypothesis

```

  note  $IH\text{-}F_1 = \text{less}(1)[OF \text{ psubset-card-mono}[OF \text{ finite-}F \text{ } F_1\text{-psubset-}F] \text{ rev-finite-subset}[OF$ 
 $\text{less}(2) \text{ } Un\text{-}F_1\text{-in-}Un\text{-}F]]$ 

```

```

  from  $\text{sh}F_0\text{-subset-sh}F \text{ } \text{sh}F_1\text{-subset-sh}F$  have  $\text{shattered-subset}$ :  $(\text{shattered-by } ?F_0) \cup (\text{shattered-by } ?F_1) \subseteq \text{shattered-by } F$  by simp

```


There is a set with the same cardinality as the intersection of *shattered-by* $\{S \in F. x \notin S\}$ and *shattered-by* $\{S \in F. x \in S\}$ which is disjoint from their union, which is also contained in *shattered-by* F .

have *f-copies-the-intersection*:
 $\exists f. \text{inj-on } f \text{ (shattered-by ?F0 } \cap \text{ shattered-by ?F1) } \wedge$
 $(\text{shattered-by ?F0 } \cup \text{ shattered-by ?F1}) \cap (f \text{ ' (shattered-by ?F0 } \cap \text{ shattered-by ?F1)}) = \{\}$ \wedge
 $f \text{ ' (shattered-by ?F0 } \cap \text{ shattered-by ?F1)} \subseteq \text{shattered-by } F$
proof
have *x-not-in-shattered*: $\forall S \in (\text{shattered-by ?F0}) \cup (\text{shattered-by ?F1}). x \notin S$
unfolding *shattered-by-def* **by** *blast*

This set is precisely the image of the intersection under *insert* x .

let $?f = \text{insert } x$
have 0 : *inj-on* $?f \text{ (shattered-by ?F0 } \cap \text{ shattered-by ?F1)}$
proof
fix $X \ Y$
assume $x0$: $X \in (\text{shattered-by ?F0 } \cap \text{ shattered-by ?F1})$ **and** $y0$: $Y \in (\text{shattered-by ?F0 } \cap \text{ shattered-by ?F1})$
and 0 : $?f \ X = ?f \ Y$
from *x-not-in-shattered* $x0$ **have** $X = ?f \ X - \{x\}$ **by** *blast*
also from 0 **have** $\dots = ?f \ Y - \{x\}$ **by** *argo*
also from *x-not-in-shattered* $y0$ **have** $\dots = Y$ **by** *blast*
finally show $X = Y$.
qed

The set is disjoint from the union.

have 1 : $(\text{shattered-by ?F0 } \cup \text{ shattered-by ?F1}) \cap ?f \text{ ' (shattered-by ?F0 } \cap \text{ shattered-by ?F1}) = \{\}$
proof (*rule ccontr*)
assume $(\text{shattered-by ?F0 } \cup \text{ shattered-by ?F1}) \cap ?f \text{ ' (shattered-by ?F0 } \cap \text{ shattered-by ?F1}) \neq \{\}$
then obtain S **where** 10 : $S \in (\text{shattered-by ?F0 } \cup \text{ shattered-by ?F1})$ **and**
 11 : $S \in ?f \text{ ' (shattered-by ?F0 } \cap \text{ shattered-by ?F1)}$ **by** *auto*
from 10 *x-not-in-shattered* **have** $x \notin S$ **by** *blast*
with 11 **show** *False* **by** *blast*
qed

This set is also in *shattered-by* F .

have 2 : $?f \text{ ' (shattered-by ?F0 } \cap \text{ shattered-by ?F1)} \subseteq \text{shattered-by } F$
proof
fix $S\text{-}x$
assume $S\text{-}x \in ?f \text{ ' (shattered-by ?F0 } \cap \text{ shattered-by ?F1)}$
then obtain S **where** 20 : $S \in \text{shattered-by ?F0}$ **and** 21 : $S \in \text{shattered-by ?F1}$ **and** 22 : $S\text{-}x = ?f \ S$ **by** *blast*
from *x-not-in-shattered* 20 **have** *x-not-in-S*: $x \notin S$ **by** *blast*
from 22 *Pow-insert*[*of* $x \ S$] **have** $\text{Pow } S\text{-}x = \text{Pow } S \cup ?f \text{ ' Pow } S$ **by** *fast*

also from 20 have $\dots = (?F0 \cap^* S) \cup (?f \text{ ' } Pow \ S)$ **unfolding shattered-by-def by blast**
also from 21 have $\dots = (?F0 \cap^* S) \cup (?f \text{ ' } (?F1 \cap^* S))$ **unfolding shattered-by-def by force**
also from insert-IntF[*of x S ?F1*] have $\dots = (?F0 \cap^* S) \cup (?f \text{ ' } ?F1 \cap^* (?f \ S))$ **by argo**
also from 22 have $\dots = (?F0 \cap^* S) \cup (?F1 \cap^* S-x)$ **by blast**
also from 22 have $\dots = (?F0 \cap^* S-x) \cup (?F1 \cap^* S-x)$ **by blast**
also from subset-IntF[*OF F0-in-F, of S-x*] subset-IntF[*OF F1-in-F, of S-x*]
have $\dots \subseteq (F \cap^* S-x)$ **by blast**
finally have $Pow \ S-x \subseteq (F \cap^* S-x)$.
thus $S-x \in \text{shattered-by } F$ **unfolding shattered-by-def by blast**
qed

from 0 1 2 show $\text{inj-on } ?f \ (\text{shattered-by } ?F0 \cap \text{shattered-by } ?F1) \wedge$
 $(\text{shattered-by } ?F0 \cup \text{shattered-by } ?F1) \cap (?f \text{ ' } (\text{shattered-by } ?F0 \cap \text{shattered-by } ?F1)) = \{\}$ **by blast**
 $?f \text{ ' } (\text{shattered-by } ?F0 \cap \text{shattered-by } ?F1) \subseteq \text{shattered-by } F$ **by blast**
qed

have $F0\text{-union-}F1\text{-is-}F: ?F0 \cup ?F1 = F$ **by fastforce**
from finite-F have $\text{finite-}F0: \text{finite } ?F0$ **and** $\text{finite-}F1: \text{finite } ?F1$ **by fastforce+**
have $\text{disjoint-}F0\text{-}F1: ?F0 \cap ?F1 = \{\}$ **by fastforce**

Thus we have the following lower bound on the cardinality of *shattered-by F*

from F0-union-F1-is-F card-Un-disjoint[*OF finite-F0 finite-F1 disjoint-F0-F1*]
have $\text{card } F = \text{card } ?F0 + \text{card } ?F1$ **by argo**
also from IH-F0
have $\dots \leq \text{card } (\text{shattered-by } ?F0) + \text{card } ?F1$ **by linarith**
also from IH-F1
have $\dots \leq \text{card } (\text{shattered-by } ?F0) + \text{card } (\text{shattered-by } ?F1)$ **by linarith**
also from card-Int-copy[*OF finite-shF shattered-subset f-copies-the-intersection*]
have $\dots \leq \text{card } (\text{shattered-by } F)$ **by argo**
finally show *?thesis* .
next

If F contains less than 2 sets, the statement follows trivially

case *False*

hence $\text{card } F = 0 \vee \text{card } F = 1$ **by force**

thus *?thesis*

proof

assume $\text{card } F = 0$

thus *?thesis* **by auto**

next

assume *asm*: $\text{card } F = 1$

hence $F\text{-not-empty}: F \neq \{\}$ **by fastforce**

from *shatters-empty[OF F-not-empty]* **have** $\{\{\}\} \subseteq \text{shattered-by } F$ **unfolding shattered-by-def by fastforce**

from *card-mono*[*OF finite-shF this*] *asm* **show** *?thesis* **by** *fastforce*
 qed
 qed
 qed

5.2 Sauer-Shelah Lemma

corollary *sauer-shelah*:

fixes $F :: 'a \text{ set set}$
 assumes *finite* $(\bigcup F)$ **and** $(\sum_{i \leq k}. \text{card } (\bigcup F) \text{ choose } i) < \text{card } F$
 shows $\exists S. (F \text{ shatters } S \wedge \text{card } S = k + 1)$
proof –
 let $?K = \{S. S \subseteq \bigcup F \wedge \text{card } S \leq k\}$
 from *finite-Pow-iff*[*of F*] *assms*(1) **have** *finite-Pow-Un*: *finite* $(\text{Pow } (\bigcup F))$ **by**
fast

from *sauer-shelah-0*[*OF assms*(1)] *assms*(2) **have** $(\sum_{i \leq k}. \text{card } (\bigcup F) \text{ choose } i) < \text{card } (\text{shattered-by } F)$ **by** *linarith*
 with *choose-row-sum-set*[*OF assms*(1), *of k*] **have** $\text{card } ?K < \text{card } (\text{shattered-by } F)$ **by** *presburger*

from *finite-diff-not-empty*[*OF finite-subset*[*OF - finite-Pow-Un*] *this*]
 obtain S **where** $S \in \text{shattered-by } F - ?K$ **by** *blast*
 then **have** *F-shatters-S*: $F \text{ shatters } S$ **and** $S \subseteq \bigcup F$ **and** $\neg(S \subseteq \bigcup F \wedge \text{card } S \leq k)$ **unfolding** *shattered-by-def* **by** *blast+*
 then **have** *card-S-ge-Suc-k*: $k + 1 \leq \text{card } S$ **by** *simp*
 from *obtain-subset-with-card-n*[*OF card-S-ge-Suc-k*] **obtain** S' **where** $\text{card } S' = k + 1$ **and** $S' \subseteq S$ **by** *blast*
 from *this*(1) *supset-shatters*[*OF this*(2) *F-shatters-S*] **show** *?thesis* **by** *blast*
 qed

5.3 Sauer-Shelah Lemma for hypergraphs

corollary *sauer-shelah-2*:

fixes $X :: 'a \text{ set set}$ **and** $S :: 'a \text{ set}$
 assumes *finite* S **and** $X \subseteq \text{Pow } S$ **and** $(\sum_{i \leq k}. \text{card } S \text{ choose } i) < \text{card } X$
 shows $\exists Y. (X \text{ shatters } Y \wedge \text{card } Y = k + 1)$
proof –
 from *assms*(2) **have** $0: \bigcup X \subseteq S$ **by** *blast*
 from *sum-mono*[*OF choose-mono*[*OF card-mono*[*OF assms*(1) 0]]] **have** $(\sum_{i \leq k}. \text{card } (\bigcup X) \text{ choose } i) \leq (\sum_{i \leq k}. \text{card } S \text{ choose } i)$ **by** *fast*
 with *sauer-shelah*[*OF finite-subset*[*OF 0 assms*(1)]] *assms*(3) **show** *?thesis* **by**
simp
 qed

5.4 Alternative statement of the Sauer-Shelah Lemma

corollary *sauer-shelah-alt*:

assumes *finite* $(\bigcup F)$ **and** $\text{VC-dim } F = k$
 shows $\text{card } F \leq (\sum_{i \leq k}. \text{card } (\bigcup F) \text{ choose } i)$

proof (rule ccontr)
assume $\neg \text{card } F \leq (\sum_{i \leq k}. \text{card } (\bigcup F) \text{ choose } i)$ **hence** $(\sum_{i \leq k}. \text{card } (\bigcup F) \text{ choose } i) < \text{card } F$ **by** *linarith*
from *sauer-shelah*[*OF* *assms*(1) *this*] **obtain** *S* **where** *F* *shatters* *S* **and** $\text{card } S = k + 1$ **by** *blast*
from *this*(1) *this*(2)[*symmetric*] **have** $k + 1 \in \{\text{card } S \mid S. F \text{ shatters } S\}$ **by** *blast*
from *cSup-upper*[*OF* *this* *bdd-above-finite*[*OF* *finite-image-set*[*OF* *finite-shattered-by*[*unfolded* *shattered-by-def*, *OF* *assms*(1)]]], *folded* *VC-dim-def*] *assms*(2) **show** *False* **by** *force*
qed
end

References

- [1] G. Kalai. Extremal combinatorics iii: Some basic theorems, Sep 2008.
- [2] N. Sauer. On the density of families of sets. *Journal of Combinatorial Theory, Series A*, 13(1):145–147, 1972.
- [3] S. Shelah. A combinatorial problem; stability and order for models and theories in infinitary languages. *Pacific Journal of Mathematics*, 41(1):247 – 261, 1972.
- [4] V. N. Vapnik and A. J. Červonenkis. The uniform convergence of frequencies of the appearance of events to their probabilities. *Teor. Veroyatnost. i Primenen.*, 16:264–279, 1971.