Sauer-Shelah Lemma

Ata Keskin

November 4, 2022

Abstract

The Sauer-Shelah Lemma is a fundemental result in extremal set theory and combinatorics, that guarentees the existence of a set T of size k which is shattered by a family of sets $\mathcal{F}$, if the cardinality of the family is greater than some bound dependent on k. A set T is said to be shattered by a family $\mathcal{F}$ if every subset of T can be obtained as an intersection of T with some set $S \in \mathcal{F}$. The Sauer-Shelah Lemma has found use in diverse fields such as computational geometry, approximation algorithms and machine learning. In this entry we formalize the notion of shattering and prove the generalized and the standard version of the Sauer-Shelah Lemma.

Contents

1 Definitions and lemmas about shattering

theory *Shattering*
  imports *Main*
begin

1.1 Intersection of a family of sets with a set

abbreviation *IntF* :: *'a set set* ⇒ *'a set* ⇒ *'a set set* (**infixl** ∩∗ *60*)
  where $F \cap* S \equiv ((\cap)\ S)$ ‘ F

lemma *idem-IntF*:
  assumes $\bigcup A \subseteq Y$
  shows $A \cap* Y = A$
proof −
  from *assms* **have** $A \subseteq A \cap* Y$ **by** *blast*
  thus *?thesis* **by** *fastforce*
qed

lemma *subset-IntF*:
  assumes $A \subseteq B$
  shows $A \cap* X \subseteq B \cap* X$
  using *assms* **by** (*rule image-mono*)

lemma *Int-IntF*: $(A \cap* Y) \cap* X = A \cap* (Y \cap X)$
proof
  show $A \cap* Y \cap* X \subseteq A \cap* (Y \cap X)$
  proof
    fix S
    assume $S \in A \cap* Y \cap* X$
    then obtain *a-y* **where** *A-Y0*: *a-y* $\in A \cap* Y$ **and** *A-Y1*: *a-y* $\cap X = S$ **by** *blast*
    from *A-Y0* **obtain** a **where** *A0*: $a \in A$ **and** *A1*: $a \cap Y =$ *a-y* **by** *blast*
    from *A-Y1 A1* **have** $a \cap (Y \cap X) = S$ **by** *fast*
    with *A0* **show** $S \in A \cap* (Y \cap X)$ **by** *blast*
  qed
next
  show $A \cap* (Y \cap X) \subseteq A \cap* Y \cap* X$
  proof
    fix S
    assume $S \in A \cap* (Y \cap X)$
    then obtain a **where** *A0*: $a \in A$ **and** *A1*: $a \cap (Y \cap X) = S$ **by** *blast*
    from *A0* **have** $a \cap Y \in A \cap* Y$ **by** *blast*
    with *A1* **show** $S \in (A \cap* Y) \cap* X$ **by** *blast*
  qed
qed

insert distributes over IntF

lemma *insert-IntF*:
  shows *insert* x ‘ $(H \cap* S) = ($*insert* x ‘ $H) \cap* ($*insert* $x\ S)$
proof
  show *insert* x ‘ $(H \cap* S) \subseteq ($*insert* x ‘ $H) \cap* ($*insert* $x\ S)$
  proof
    fix *y-x*
    assume *y-x* $\in$ *insert* x ‘ $(H \cap* S)$
    then obtain y **where** *0*: $y \in (H \cap* S)$ **and** *1*: *y-x* $= y \cup \{x\}$ **by** *blast*

from *0* **obtain** *yh* **where** *2*: $yh \in H$ **and** *3*: $y = yh \cap S$ **by** *blast*
from *1 3* **have** $y\text{-}x = (yh \cup \{x\}) \cap (S \cup \{x\})$ **by** *simp*
with *2* **show** $y\text{-}x \in (insert\ x\ `\ H) \cap\!* (insert\ x\ S)$ **by** *blast*
qed
next
show $insert\ x\ `\ H \cap\!* (insert\ x\ S) \subseteq insert\ x\ `\ (H \cap\!* S)$
proof
fix *y-x*
assume $y\text{-}x \in insert\ x\ `\ H \cap\!* (insert\ x\ S)$
then obtain *yh-x* **where** *0*: $yh\text{-}x \in (\lambda Y.\ Y \cup \{x\})\ `\ H$ **and** *1*: $y\text{-}x = yh\text{-}x \cap (S \cup \{x\})$ **by** *blast*
from *0* **obtain** *yh* **where** *2*: $yh \in H$ **and** *3*: $yh\text{-}x = yh \cup \{x\}$ **by** *blast*
from *1 3* **have** $y\text{-}x = (yh \cap S) \cup \{x\}$ **by** *simp*
with *2* **show** $y\text{-}x \in insert\ x\ `\ (H \cap\!* S)$ **by** *blast*
qed
qed

1.2 Definition of shattering

abbreviation *shatters* :: $'a\ set\ set \Rightarrow 'a\ set \Rightarrow bool$ (**infixl** *shatters 70*)
where $H\ shatters\ A \equiv H \cap\!* A = Pow\ A$

definition *VC-dim* :: $'a\ set\ set \Rightarrow nat$
where $VC\text{-}dim\ F = Sup\ \{card\ S \mid S.\ F\ shatters\ S\}$

definition *shattered-by* :: $'a\ set\ set \Rightarrow 'a\ set\ set$
where $shattered\text{-}by\ F \equiv \{A.\ F\ shatters\ A\}$

lemma *shattered-by-in-Pow*:
shows $shattered\text{-}by\ F \subseteq Pow\ (\bigcup\ F)$
unfolding *shattered-by-def* **by** *blast*

lemma *subset-shatters*:
assumes $A \subseteq B$ **and** $A\ shatters\ X$
shows $B\ shatters\ X$
proof −
from *assms(1)* **have** $A \cap\!* X \subseteq B \cap\!* X$ **by** *blast*
with *assms(2)* **have** $Pow\ X \subseteq B \cap\!* X$ **by** *presburger*
thus *?thesis* **by** *blast*
qed

lemma *supset-shatters*:
assumes $Y \subseteq X$ **and** $A\ shatters\ X$
shows $A\ shatters\ Y$
proof −
have *h*: $\bigcup(Pow\ Y) \subseteq Y$ **by** *simp*
from *assms* **have** *0*: $Pow\ Y \subseteq A \cap\!* X$ **by** *auto*
from *subset-IntF[OF 0, of Y] Int-IntF[of Y X A] idem-IntF[OF h]* **have** $Pow\ Y \subseteq A \cap\!* (X \cap Y)$ **by** *argo*

with *Int-absorb2*[*OF assms*(*1*)] *Int-commute*[*of X Y*] **have** *Pow Y* ⊆ *A* ∩∗ *Y*
by *presburger*
then show *?thesis* **by** *fast*
qed

lemma *shatters-empty*:
assumes *F* ≠ {}
shows *F shatters* {}
using *assms* **by** *fastforce*

lemma *subset-shattered-by*:
assumes *A* ⊆ *B*
shows *shattered-by A* ⊆ *shattered-by B*
unfolding *shattered-by-def* **using** *subset-shatters*[*OF assms*] **by** *force*

lemma *finite-shattered-by*:
assumes *finite* (⋃ *F*)
shows *finite* (*shattered-by F*)
using *assms rev-finite-subset*[*OF - shattered-by-in-Pow*, *of F*] **by** *fast*

The following example shows that requiring finiteness of a family of sets is not enough

lemma ∃ *F*::*nat set set*. *finite F* ∧ *infinite* (*shattered-by F*)
proof −
let *?F* = {*odd* −‘ {*True*}, *odd* −‘ {*False*}}
have *0*: *finite ?F* **by** *simp*

let *?f* = λ*n*::*nat*. {*n*}
let *?N* = *range ?f*
have *inj* (λ*n*. {*n*}) **by** *simp*
with *infinite-iff-countable-subset*[*of ?N*] **have** *infinite-N*: *infinite ?N* **by** *blast*
have *F-shatters-any-singleton*: *?F shatters* {*n*::*nat*} **for** *n*
proof −
have *Pow-n*: *Pow* {*n*} = {{*n*}, {}} **by** *blast*
have *1*: *Pow* {*n*} ⊆ *?F* ∩∗ {*n*}
proof (*cases odd n*)
case *True*
from *True* **have** (*odd* −‘ {*False*}) ∩ {*n*} = {} **by** *blast*
hence *0*: {} ∈ *?F* ∩∗ {*n*} **by** *blast*
from *True* **have** (*odd* −‘ {*True*}) ∩ {*n*} = {*n*} **by** *blast*
hence *1*: {*n*} ∈ *?F* ∩∗ {*n*} **by** *blast*
from *0 1 Pow-n* **show** *?thesis* **by** *simp*
next
case *False*
from *False* **have** (*odd* −‘ {*True*}) ∩ {*n*} = {} **by** *blast*
hence *0*: {} ∈ *?F* ∩∗ {*n*} **by** *blast*
from *False* **have** (*odd* −‘ {*False*}) ∩ {*n*} = {*n*} **by** *blast*
hence *1*: {*n*} ∈ *?F* ∩∗ {*n*} **by** *blast*
from *0 1 Pow-n* **show** *?thesis* **by** *simp*

qed
thus *?thesis* **by** *fastforce*
qed
then have *?N* $\subseteq$ *shattered-by ?F* **unfolding** *shattered-by-def* **by** *force*
from *0 infinite-super*[*OF this infinite-N*] **show** *?thesis* **by** *blast*
qed

end

2 Lemmas involving the cardinality of sets

theory *Card-Lemmas*
imports *Main*
begin

lemma *card-diff*:
assumes *finite A*
shows *card A* = *card* (*A* − *B*) + *card* (*A* ∩ *B*)
proof −
from *assms* **have** *fin0*: *finite* (*A* − *B*) **and** *fin1*: *finite* (*A* ∩ *B*) **by** *blast+*
have *A-equ*: *A* = (*A* − *B*) ∪ (*A* ∩ *B*) **and** *disjoint*: (*A* − *B*) ∩ (*A* ∩ *B*) = {} **by** *blast+*
from *card-Un-disjoint*[*OF fin0 fin1 disjoint*] *A-equ* **show** *?thesis* **by** *argo*
qed

lemma *card-Int-copy*:
assumes *finite X* **and** *A* ∪ *B* ⊆ *X* **and** ∃*f*. *inj-on f* (*A* ∩ *B*) ∧ (*A* ∪ *B*) ∩ (*f* ‘ (*A* ∩ *B*)) = {} ∧ *f* ‘ (*A* ∩ *B*) ⊆ *X*
shows *card A* + *card B* ≤ *card X*
proof −
from *rev-finite-subset*[*OF assms*(*1*), *of A*] *rev-finite-subset*[*OF assms*(*1*), *of B*] *assms*(*2*)
have *finite-A*: *finite A* **and** *finite-B*: *finite B* **by** *blast+*
then have *finite-A-Un-B*: *finite* (*A* ∪ *B*) **and** *finite-A-Int-B*: *finite* (*A* ∩ *B*) **by** *blast+*
from *assms*(*3*) **obtain** *f* **where** *f-inj-on*: *inj-on f* (*A* ∩ *B*) **and** *f-disjnt*: (*A* ∪ *B*) ∩ (*f* ‘ (*A* ∩ *B*)) = {} **and** *f-imj-in*: *f* ‘ (*A* ∩ *B*) ⊆ *X* **by** *blast*
from *finite-A-Int-B* **have** *finite-f-img*: *finite* (*f* ‘ (*A* ∩ *B*)) **by** *blast*
from *assms*(*2*) *f-imj-in* **have** *union-in*: (*A* ∪ *B*) ∪ *f* ‘ (*A* ∩ *B*) ⊆ *X* **by** *blast*

from *card-Un-Int*[*OF finite-A finite-B*] **have** *card A* + *card B* = *card* (*A* ∪ *B*) + *card* (*A* ∩ *B*) **.**
also from *card-image*[*OF f-inj-on*] **have** ... = *card* (*A* ∪ *B*) + *card* (*f* ‘ (*A* ∩ *B*)) **by** *presburger*
also from *card-Un-disjoint*[*OF finite-A-Un-B finite-f-img f-disjnt*] **have** ... = *card* ((*A* ∪ *B*) ∪ *f* ‘ (*A* ∩ *B*)) **by** *argo*
also from *card-mono*[*OF assms*(*1*) *union-in*] **have** ... ≤ *card X* **by** *blast*
finally show *?thesis* **.**

qed

lemma *card-ge-0*:
  assumes $A \neq \{\}$ **and** *finite A*
  shows $0 < card\ A$
proof −
  from *assms*(*1*) **have** $\{\} \subset A$ **by** *blast*
  from *psubset-card-mono*[*OF assms*(*2*) *this*] **show** *?thesis* **by** *force*
qed

lemma *finite-diff-not-empty*:
  assumes *finite Y* **and** $card\ Y < card\ X$
  shows $X - Y \neq \{\}$
proof
  assume $X - Y = \{\}$
  hence $X \subseteq Y$ **by** *simp*
  from *card-mono*[*OF assms*(*1*) *this*] *assms*(*2*) **show** *False* **by** *linarith*
qed

lemma *obtain-difference-element*:
  fixes F :: *'a set set*
  assumes $2 \leq card\ F$
  obtains x **where** $x \in \bigcup F$ $x \notin \bigcap F$
proof −
  from *assms card-le-Suc-iff*[*of 1 F*] **obtain** $A\ F'$ **where** *0*: $F = insert\ A\ F'$ **and** *1*: $A \notin F'$ **and** *2*: $1 \leq card\ F'$ **by** *auto*
  from *2 card-le-Suc-iff*[*of 0 F'*] **obtain** $B\ F''$ **where** *3*: $F' = insert\ B\ F''$ **by** *auto*
  from *1 3* **have** *A-noteq-B*: $A \neq B$ **by** *blast*
  from *0 3* **have** *A-in-F*: $A \in F$ **and** *B-in-F*: $B \in F$ **by** *blast+*
  from *A-noteq-B* **have** $(A - B) \cup (B - A) \neq \{\}$ **by** *simp*
  with *A-in-F B-in-F that* **show** *thesis* **by** *blast*
qed

end

3 Lemmas involving the binomial coefficient

theory *Binomial-Lemmas*
  imports *Main*
begin

lemma *choose-mono*:
  assumes $x \leq y$
  shows $x\ choose\ n \leq y\ choose\ n$
proof −
  have *finite* $\{0..<y\}$ **by** *blast*
  with *finite-Pow-iff*[*of* $\{0..<y\}$] **have** *finiteness*: *finite* $\{K \in Pow\ \{0..<y\}.\ card\ K = n\}$ **by** *simp*

from *assms* **have** $Pow\ \{0..<x\} \subseteq Pow\ \{0..<y\}$ **by** *force*
then have $\{K \in Pow\ \{0..<x\}.\ card\ K = n\} \subseteq \{K \in Pow\ \{0..<y\}.\ card\ K = n\}$ **by** *blast*
from *card-mono*[*OF finiteness this*] **show** *?thesis* **unfolding** *binomial-def* **.**
qed

lemma *choose-row-sum-set*:
assumes *finite* $(\bigcup F)$
shows $card\ \{S.\ S \subseteq \bigcup F \wedge card\ S \leq k\} = (\sum i\leq k.\ card\ (\bigcup\ F)\ choose\ i)$
proof (*induction k*)
case *0*
from *rev-finite-subset*[*OF assms*] **have** $S \subseteq \bigcup F \wedge card\ S \leq 0 \longleftrightarrow S = \{\}$ **for** S **by** *fastforce*
then show *?case* **by** *simp*
next
case (*Suc k*)
let $?FS = \{S.\ S \subseteq \bigcup\ F \wedge card\ S \leq Suc\ k\}$
and $?F\text{-}Asm = \{S.\ S \subseteq \bigcup\ F \wedge card\ S \leq k\}$
and $?F\text{-}Step = \{S.\ S \subseteq \bigcup\ F \wedge card\ S = Suc\ k\}$

from *finite-Pow-iff*[*of* $\bigcup F$] *assms* **have** *finite-Pow-Un-F*: *finite* $(Pow\ (\bigcup\ F))$ **..**
have $?F\text{-}Asm \subseteq Pow\ (\bigcup\ F)$ **and** $?F\text{-}Step \subseteq Pow\ (\bigcup\ F)$ **by** *fast+*
with *rev-finite-subset*[*OF finite-Pow-Un-F*] **have** *finite-F-Asm*: *finite ?F-Asm* **and** *finite-F-Step*: *finite ?F-Step* **by** *presburger+*

have *F-Un*: $?FS = ?F\text{-}Asm \cup ?F\text{-}Step$ **and** *F-disjoint*: $?F\text{-}Asm \cap ?F\text{-}Step = \{\}$ **by** *fastforce+*
from *card-Un-disjoint*[*OF finite-F-Asm finite-F-Step F-disjoint*] *F-Un* **have** $card\ ?FS = card\ ?F\text{-}Asm + card\ ?F\text{-}Step$ **by** *argo*
also from *Suc* **have** $... = (\sum i\leq k.\ card\ (\bigcup\ F)\ choose\ i) + card\ ?F\text{-}Step$ **by** *argo*
also from *n-subsets*[*OF assms*, *of Suc k*] **have** $... = (\sum i\leq Suc\ k.\ card\ (\bigcup\ F)\ choose\ i)$ **by** *force*
finally show *?case* **by** *blast*
qed

end

4 Sauer-Shelah Lemma

theory *Sauer-Shelah-Lemma*
imports *Main Shattering Card-Lemmas Binomial-Lemmas*
begin

4.1 Generalized Sauer-Shelah Lemma

lemma *sauer-shelah-0*:
fixes $F :: 'a\ set\ set$
shows $finite\ (\bigcup\ F) \Longrightarrow card\ F \leq card\ (shattered\text{-}by\ F)$
proof (*induction F rule*: *measure-induct-rule*[*of card*])

case $(less\ F)$
note $finite\text{-}F = finite\text{-}UnionD[OF\ less(2)]$
note $finite\text{-}shF = finite\text{-}shattered\text{-}by[OF\ less(2)]$
show $?case$
proof $(cases\ 2 \le card\ F)$
case $True$
from $obtain\text{-}difference\text{-}element[OF\ True]$ **obtain** $x :: 'a$ **where** $x\text{-}in\text{-}Union\text{-}F$: $x \in \bigcup F$ **and** $x\text{-}not\text{-}in\text{-}Int\text{-}F$: $x \notin \bigcap F$ **by** $blast$

Define F0 as the subfamily of F containing those sets that don't contain x

let $?F0 = \{S \in F.\ x \notin S\}$
from $x\text{-}in\text{-}Union\text{-}F$ **have** $F0\text{-}psubset\text{-}F$: $?F0 \subset F$ **by** $blast$
from $F0\text{-}psubset\text{-}F$ **have** $F0\text{-}in\text{-}F$: $?F0 \subseteq F$ **by** $blast$
from $subset\text{-}shattered\text{-}by[OF\ F0\text{-}in\text{-}F]$ **have** $shF0\text{-}subset\text{-}shF$: $shattered\text{-}by\ ?F0 \subseteq shattered\text{-}by\ F$ **.**
from $F0\text{-}in\text{-}F$ **have** $Un\text{-}F0\text{-}in\text{-}Un\text{-}F$:$\bigcup\ ?F0 \subseteq \bigcup\ F$ **by** $blast$

F0 shatters at least as many sets as |F0| by the induction hypothesis

note $IH\text{-}F0 = less(1)[OF\ psubset\text{-}card\text{-}mono[OF\ finite\text{-}F\ F0\text{-}psubset\text{-}F]\ rev\text{-}finite\text{-}subset[OF\ less(2)\ Un\text{-}F0\text{-}in\text{-}Un\text{-}F]]$

Define F1 as the subfamily of F containing those sets that contain x

let $?F1 = \{S \in F.\ x \in S\}$
from $x\text{-}not\text{-}in\text{-}Int\text{-}F$ **have** $F1\text{-}psubset\text{-}F$: $?F1 \subset F$ **by** $blast$
from $F1\text{-}psubset\text{-}F$ **have** $F1\text{-}in\text{-}F$: $?F1 \subseteq F$ **by** $blast$
from $subset\text{-}shattered\text{-}by[OF\ F1\text{-}in\text{-}F]$ **have** $shF1\text{-}subset\text{-}shF$: $shattered\text{-}by\ ?F1 \subseteq shattered\text{-}by\ F$ **.**
from $F1\text{-}in\text{-}F$ **have** $Un\text{-}F1\text{-}in\text{-}Un\text{-}F$:$\bigcup\ ?F1 \subseteq \bigcup\ F$ **by** $blast$

F1 shatters at least as many sets as |F1| by the induction hypothesis

note $IH\text{-}F1 = less(1)[OF\ psubset\text{-}card\text{-}mono[OF\ finite\text{-}F\ F1\text{-}psubset\text{-}F]\ rev\text{-}finite\text{-}subset[OF\ less(2)\ Un\text{-}F1\text{-}in\text{-}Un\text{-}F]]$

from $shF0\text{-}subset\text{-}shF\ shF1\text{-}subset\text{-}shF$ **have** $shattered\text{-}subset$: $(shattered\text{-}by\ ?F0) \cup (shattered\text{-}by\ ?F1) \subseteq shattered\text{-}by\ F$ **by** $simp$

There is a set with the same cardinality as the intersection of *shattered-by* $\{S \in F.\ x \notin S\}$ and *shattered-by* $\{S \in F.\ x \in S\}$ which is disjoint from their union, which is also contained in *shattered-by F*.

have $f\text{-}copies\text{-}the\text{-}intersection$:
$\exists f.\ inj\text{-}on\ f\ (shattered\text{-}by\ ?F0 \cap shattered\text{-}by\ ?F1) \wedge$
$(shattered\text{-}by\ ?F0 \cup shattered\text{-}by\ ?F1) \cap (f\ `\ (shattered\text{-}by\ ?F0 \cap shattered\text{-}by\ ?F1)) = \{\} \wedge$
$f\ `\ (shattered\text{-}by\ ?F0 \cap shattered\text{-}by\ ?F1) \subseteq shattered\text{-}by\ F$
proof
have $x\text{-}not\text{-}in\text{-}shattered$: $\forall S \in (shattered\text{-}by\ ?F0) \cup (shattered\text{-}by\ ?F1).\ x \notin S$ **unfolding** $shattered\text{-}by\text{-}def$ **by** $blast$

This set is precisely the image of the intersection under *insert x*.

let *?f* = *insert x*
have *0*: *inj-on ?f* (*shattered-by ?F0* ∩ *shattered-by ?F1*)
proof
fix *X Y*
assume *x0*: *X* ∈ (*shattered-by ?F0* ∩ *shattered-by ?F1*) **and** *y0*: *Y* ∈ (*shattered-by ?F0* ∩ *shattered-by ?F1*)
and *0*: *?f X* = *?f Y*
from *x-not-in-shattered x0* **have** *X* = *?f X* − {*x*} **by** *blast*
also from *0* **have** ... = *?f Y* − {*x*} **by** *argo*
also from *x-not-in-shattered y0* **have** ... = *Y* **by** *blast*
finally show *X* = *Y* **.**
qed

The set is disjoint from the union.

have *1*: (*shattered-by ?F0* ∪ *shattered-by ?F1*) ∩ *?f* ‘ (*shattered-by ?F0* ∩ *shattered-by ?F1*) = {}
proof (*rule ccontr*)
assume (*shattered-by ?F0* ∪ *shattered-by ?F1*) ∩ *?f* ‘ (*shattered-by ?F0* ∩ *shattered-by ?F1*) ≠ {}
then obtain *S* **where** *10*: *S* ∈ (*shattered-by ?F0* ∪ *shattered-by ?F1*) **and** *11*: *S* ∈ *?f* ‘ (*shattered-by ?F0* ∩ *shattered-by ?F1*) **by** *auto*
from *10 x-not-in-shattered* **have** *x* ∉ *S* **by** *blast*
with *11* **show** *False* **by** *blast*
qed

This set is also in *shattered-by F*.

have *2*: *?f* ‘ (*shattered-by ?F0* ∩ *shattered-by ?F1*) ⊆ *shattered-by F*
proof
fix *S-x*
assume *S-x* ∈ *?f* ‘ (*shattered-by ?F0* ∩ *shattered-by ?F1*)
then obtain *S* **where** *20*: *S* ∈ *shattered-by ?F0* **and** *21*: *S* ∈ *shattered-by ?F1* **and** *22*: *S-x* = *?f S* **by** *blast*
from *x-not-in-shattered 20* **have** *x-not-in-S*: *x* ∉ *S* **by** *blast*

from *22 Pow-insert*[*of x S*] **have** *Pow S-x* = *Pow S* ∪ *?f* ‘ *Pow S* **by** *fast*
also from *20* **have** ... = (*?F0* ∩∗ *S*) ∪ (*?f* ‘ *Pow S*) **unfolding** *shattered-by-def* **by** *blast*
also from *21* **have** ... = (*?F0* ∩∗ *S*) ∪ (*?f* ‘ (*?F1* ∩∗ *S*)) **unfolding** *shattered-by-def* **by** *force*
also from *insert-IntF*[*of x S ?F1*] **have** ... = (*?F0* ∩∗ *S*) ∪ (*?f* ‘ *?F1* ∩∗ (*?f S*)) **by** *argo*
also from *22* **have** ... = (*?F0* ∩∗ *S*) ∪ (*?F1* ∩∗ *S-x*) **by** *blast*
also from *22* **have** ... = (*?F0* ∩∗ *S-x*) ∪ (*?F1* ∩∗ *S-x*) **by** *blast*
also from *subset-IntF*[*OF F0-in-F*, *of S-x*] *subset-IntF*[*OF F1-in-F*, *of S-x*] **have** ... ⊆ (*F* ∩∗ *S-x*) **by** *blast*
finally have *Pow S-x* ⊆ (*F* ∩∗ *S-x*) **.**
thus *S-x* ∈ *shattered-by F* **unfolding** *shattered-by-def* **by** *blast*
qed

from *0 1 2* **show** *inj-on ?f* (*shattered-by ?F0* ∩ *shattered-by ?F1*) ∧
(*shattered-by ?F0* ∪ *shattered-by ?F1*) ∩ (*?f* ‘ (*shattered-by ?F0* ∩ *shattered-by ?F1*)) = {} ∧
?f ‘ (*shattered-by ?F0* ∩ *shattered-by ?F1*) ⊆ *shattered-by F* **by** *blast*
qed

have *F0-union-F1-is-F*: *?F0* ∪ *?F1* = *F* **by** *fastforce*
from *finite-F* **have** *finite-F0*: *finite ?F0* **and** *finite-F1*: *finite ?F1* **by** *fastforce+*
have *disjoint-F0-F1*: *?F0* ∩ *?F1* = {} **by** *fastforce*

Thus we have the following lower bound on the cardinality of *shattered-by F*

from *F0-union-F1-is-F card-Un-disjoint*[*OF finite-F0 finite-F1 disjoint-F0-F1*]

have *card F* = *card ?F0* + *card ?F1* **by** *argo*
also from *IH-F0*
have ... ≤ *card* (*shattered-by ?F0*) + *card ?F1* **by** *linarith*
also from *IH-F1*
have ... ≤ *card* (*shattered-by ?F0*) + *card* (*shattered-by ?F1*) **by** *linarith*
also from *card-Int-copy*[*OF finite-shF shattered-subset f-copies-the-intersection*]
have ... ≤ *card* (*shattered-by F*) **by** *argo*
finally show *?thesis* **.**
next

If F contains less than 2 sets, the statement follows trivially

case *False*
hence *card F* = *0* ∨ *card F* = *1* **by** *force*
thus *?thesis*
proof
assume *card F* = *0*
thus *?thesis* **by** *auto*
next
assume *asm*: *card F* = *1*
hence *F-not-empty*: *F* ≠ {} **by** *fastforce*
from *shatters-empty*[*OF F-not-empty*] **have** {{}} ⊆ *shattered-by F* **unfolding** *shattered-by-def* **by** *fastforce*
from *card-mono*[*OF finite-shF this*] *asm* **show** *?thesis* **by** *fastforce*
qed
qed
qed

4.2 Sauer-Shelah Lemma

corollary *sauer-shelah*:
fixes *F* :: *'a set set*
assumes *finite* (⋃*F*) **and** (∑ *i*≤*k*. *card* (⋃*F*) *choose i*) < *card F*
shows ∃ *S*. (*F shatters S* ∧ *card S* = *k* + *1*)
proof −

let *?K* = {*S*. *S* ⊆ ⋃*F* ∧ *card S* ≤ *k*}
from *finite-Pow-iff*[*of F*] *assms*(*1*) **have** *finite-Pow-Un*: *finite* (*Pow* (⋃*F*)) **by** *fast*

from *sauer-shelah-0*[*OF assms*(*1*)] *assms*(*2*) **have** (∑*i*≤*k*. *card* (⋃*F*) *choose i*) < *card* (*shattered-by F*) **by** *linarith*
with *choose-row-sum-set*[*OF assms*(*1*), *of k*] **have** *card ?K* < *card* (*shattered-by F*) **by** *presburger*

from *finite-diff-not-empty*[*OF finite-subset*[*OF - finite-Pow-Un*] *this*]
obtain *S* **where** *S* ∈ *shattered-by F* − *?K* **by** *blast*
then have *F-shatters-S*: *F shatters S* **and** *S* ⊆ ⋃*F* **and** ¬(*S* ⊆ ⋃*F* ∧ *card S* ≤ *k*) **unfolding** *shattered-by-def* **by** *blast*+
then have *card-S-ge-Suc-k*: *k* + *1* ≤ *card S* **by** *simp*
from *obtain-subset-with-card-n*[*OF card-S-ge-Suc-k*] **obtain** *S'* **where** *card S'* = *k* + *1* **and** *S'* ⊆ *S* **by** *blast*
from *this*(*1*) *supset-shatters*[*OF this*(*2*) *F-shatters-S*] **show** *?thesis* **by** *blast*
qed

4.3 Sauer-Shelah Lemma for hypergraphs

corollary *sauer-shelah-2*:
fixes *X* :: *'a set set* **and** *S* :: *'a set*
assumes *finite S* **and** *X* ⊆ *Pow S* **and** (∑*i*≤*k*. *card S choose i*) < *card X*
shows ∃ *Y*. (*X shatters Y* ∧ *card Y* = *k* + *1*)
proof −
from *assms*(*2*) **have** *0*: ⋃*X* ⊆ *S* **by** *blast*
from *sum-mono*[*OF choose-mono*[*OF card-mono*[*OF assms*(*1*) *0*]]] **have** (∑*i*≤*k*. *card* (⋃*X*) *choose i*) ≤ (∑*i*≤*k*. *card S choose i*) **by** *fast*
with *sauer-shelah*[*OF finite-subset*[*OF 0 assms*(*1*)]] *assms*(*3*) **show** *?thesis* **by** *simp*
qed

4.4 Alternative statement of the Sauer-Shelah Lemma

corollary *sauer-shelah-alt*:
assumes *finite* (⋃*F*) **and** *VC-dim F* = *k*
shows *card F* ≤ (∑*i*≤*k*. *card* (⋃*F*) *choose i*)
proof (*rule ccontr*)
assume ¬ *card F* ≤ (∑*i*≤*k*. *card* (⋃*F*) *choose i*) **hence** (∑*i*≤*k*. *card* (⋃*F*) *choose i*) < *card F* **by** *linarith*
from *sauer-shelah*[*OF assms*(*1*) *this*] **obtain** *S* **where** *F shatters S* **and** *card S* = *k* + *1* **by** *blast*
from *this*(*1*) *this*(*2*)[*symmetric*] **have** *k* + *1* ∈ {*card S* | *S*. *F shatters S*} **by** *blast*
from *cSup-upper*[*OF this bdd-above-finite*[*OF finite-image-set*[*OF finite-shattered-by*[*unfolded shattered-by-def*, *OF assms*(*1*)]]], *folded VC-dim-def*]
assms(*2*) **show** *False* **by** *force*
qed

end